Initial Thoughts and Statement

The Lebesgue dominated convergence theorem (LDCT) is a classical result in measure theory has wide applications to many further topics. A possible statement of the theorem (given in the lecture notes for MATH50006 Measure Theory and Integration made by Dr. Pierre-François Rodriguez) is

Theorem 1 (Lebesgue Dominated Convergence Theorem)**.** *Let $g : X \to [0, \infty]$ be integrable, i.e., $g \in L^1(\mu)$, and let $f, f_n : X \to [-\infty, \infty]$, $n \geq 1$ be measurable functions such that*

$$|f_n(x)| \leq g(x) \quad \text{for all } x \in X, \quad \text{and} \quad f_n \xrightarrow{n\to\infty} f \quad \mu\text{-a.e.}$$

Then,

$$\int |f_n - f| \, d\mu \xrightarrow{n\to\infty} 0.$$

To given a formal statement of this in lean, we need to first find ways to instantiate a measure space, represent an integral, impose the conditions of measurability and integrability and encode pointwise convergence μ almost everywhere. The first statement I concocted of the theorem is

```
variable {X : Type} [MeasurableSpace X] {μ : Measure X}
  {f : X → EReal} {fₙ : ℕ → X → EReal} {g : X → ENNReal}

theorem LDCT
    (hg_int : Integrable g μ) -- integrability of g
    (hf_meas : Measurable f) -- measurability of f
    (hfn_meas : ∀ n, Measurable (fₙ n)) -- measurability of fₙ
    (hfn_to_f : ∀ᵐ (x : X) ∂μ, Tendsto (fun n ↦ fₙ n x) atTop (nhds (f x)))
    (hfn_bound : ∀ n, ∀ x, (fₙ n x).abs ≤ g x) -- dominated by g
    (hf_bound : ∀ x, (f x).abs ≤ g x) : -- f is dominated by g
    (Tendsto (fun n ↦ ∫⁻ x, (fₙ n x - f x).abs ∂μ) atTop (nhds 0))
```

However, the error `failed to synthesize ENorm EReal` was raised. This error was raised because the extended reals does not have an in built norm. This was fixed by creating an instance of ENorm over EReal using `EReal.abs`. Furthermore, initially, I used ∫, which is the Bochner integral which requires the functions being integrated to have a Banach space as its co-domain. However, as the integral in the output of the theorem is for `(fₙ n x - f x).abs` which is in `ENNReal`, we can use the lower Lebesgue integral implementation `∫⁻` for `NNReal` functions.

Strategy and Explanation of Formal Proof

The informal proof given in the lecture notes is

Proof. We have $f_n \leq g$ and $f_n \to f$ μ-a.e. on X, hence $f_n, f \in L^1(\mu)$ by monotonicity of the lebesgue integral. Moreover,$|f_n - f| \leq |f_n| + |f| \leq 2\,g$, thus $f_n - f \in L^1(\mu)$. Next, we apply Fatou's Lemma to the sequence of nonnegative functions $2\,g - |f_n - f| \geq 0$, which yields

$$\int 2g \, d\mu = \int \liminf_{n\to\infty} \left(2g - |f_n - f|\right) d\mu \leq \liminf_{n\to\infty} \int \left(2g - |f_n - f|\right) d\mu = \int 2g \, d\mu + \liminf_{n\to\infty} \int -|f_n - f| \, d\mu$$

Rearranging shows $\limsup_{n\to\infty} \int |f_n - f| \, d\mu \leq 0$ and as the sequence is bounded below we have $\lim_{n\to\infty} \int |f_n - f| \, d\mu = 0$.. □

Looking at this proof we see there is one "big" calculation involved in showing that the limsup is bounded above by 0, which I expect to be the "meat" of the proof.

Integrability of f and f_n and the `bounded_by_integrable` lemma

As the first two results we want are the integrability of f and f_n, I decided to implement the following lemma

```
lemma bounded_by_integrable (hf_meas : Measurable f) (hg_int : Integrable g μ)
(hf_bound : ∀ x, (f x).abs ≤ g x) : Integrable f μ
```

Informally, the proof for this would use the monoticity of the Lebesgue integral to show that $\int |f| d\mu \leq \int g \, d\mu < \infty$, implying f is integrable. Additionally, in Mathlib, to show that something is integrable it requires proofs of AEStrongMeasurablility and finite integral of norm. Measurability of a function is equivalent to AEStrongMeasurability if we are in a PseudoMetrizableSpace which EReal is. Therefore, we can extract the strong measurability using the theorem `aestronglyMeasurable` to get the first required proof. Next, we use monotonicity of the lower Lebesgue integral to get the finiteness of normed f. It is interesting to note that there is a lemma in Mathlib, `MeasureTheory.Integrable.mono'`, which gives this more directly but the fact that the co-domain of f is `EReals`, which is not a Banach space, seems to disqualify us from using this lemma. This would be the start of the issues with compatibility with `EReals`, `ENNReals` and reals. Originally, this was to be included in the statement of the LDCT as it is given in the lecture notes, but I have since removed it from the LDCT because it was not actually required by LEAN to close the main goal of LDCT. However, I included this lemma because I thought it could be useful and adds a new API.

Convergence of $\int |f_n - f| d\mu$ to 0

As in the informal proof, the main strategy would be to show that 0 bounds the liminf and limsup and thus the sequence of integrals must converge to 0. The lemma I found in Mathlib was `tendsto_of_le_liminf_of_limsup_le`. This lemma takes proofs that the sequence of integrals is bounded, liminf is bounded below and limsup is bounded above. Thus, I created 4 `have` statements and closed 3 pretty simply as follows:

1. The sequence of integrals is bounded below by 0 and this was closed using `filter_upwards` and `simp`.
2. Likewise for the upper bound by ∞.
3. The lower bound of liminf by 0 is trivially closed by `simp`.

Now, all that is left to show that limsup is bounded above by 0.

limsup bounded above by 0

Initially, I thought I could have this within the proof itself but it quickly started running away from me so I made an auxilliary lemma,

```
lemma limsup_aux
(hg_int : Integrable g μ) (hf_meas : Measurable f) (hfn_meas : ∀ n, Measurable (fₙ n))
(hf_int : Integrable f μ) (hfn_int : ∀ n, Integrable (fₙ n) μ)
(hfn_to_f : ∀ᵐ (x : X) ∂μ, Tendsto (fun n ↦ fₙ n x - f x) atTop (nhds 0))
(hfn_bound : ∀ n, ∀ x, (fₙ n x).abs ≤ g x) (hf_bound : ∀ x, (f x).abs ≤ g x) :
(limsup (fun n ↦ (∫⁻ x, (fₙ n x - f x).abs ∂μ)) atTop) ≤ 0 := by
```

The main strategy in proving this is an application of Fatous's lemma for lim sup and this is given in Mathlib as `limsup_lintegral_le`. The calc block which does this is

```
calc
    limsup (fun n ↦ (∫⁻ x, (fₙ n x - f x).abs ∂μ)) atTop
    ≤ ∫⁻ x, limsup (fun n ↦ (fₙ n x - f x).abs) atTop ∂μ :=
      @limsup_lintegral_le X _ μ (fun n ↦ (fun x ↦ (fₙ n x - f x).abs)) (2 * g)
        (h_meas) (hffn_le_2g) (h_2g_finite)
    _ = ∫⁻ x, 0 ∂μ := lintegral_congr_ae h_swap
    _ = 0 := by simp
```

In this, the first inequality is an application of the mentioned Fatou's lemma. The second inequality is an application of `lintegral_congr_a` which says that the integral of functions which are equal almost everywhere are equal and the proof it eats, `h_swap` was implemented in a have statement as below

```
have h_swap : (fun x ↦ limsup (fun n ↦ (fₙ n x - f x).abs) atTop) =ᵐ[μ] 0 := by
    filter_upwards [hfn_to_f] with x hfn_to_f
    apply tendsto_EReal_abs at hfn_to_f
    apply Tendsto.limsup_eq at hfn_to_f
    simpa only [Pi.zero_apply]
```

In this have, I made use of another lemma which should be in Mathlib which is that if a sequence in the EReal converges to 0 then the absolute value of 0, unfortunately, I could not prove this in time so I left it as a sorry. I got really close to it and the only issue was that I had the statement for convergence in the `EReal`s but I could not figure out how to use this to get convergence for the exact same sequence embedded in `ENNReal`s.
In pursuit of this proof, some other lemmas were implemented for `EReals`, namely

```
lemma abs_ne_top_iff {x : EReal} : x.abs ≠ ⊤ ↔ x ≠ ⊤ ∧ x ≠ ⊥

lemma EReal_abs_add_le_add {x y : EReal} : (x + y).abs ≤ x.abs + y.abs
```

```
lemma EReal_abs_sub_le_add (x y : EReal) : (x - y).abs ≤ x.abs + y.abs

lemma EReal_neg_abs_le (x : EReal) : -x ≤ x.abs

lemma EReal_le_abs_self (x : EReal) : x.abs ≤ -x ∨ x.abs ≤ x

lemma EReal_abs_le_max (x : EReal) : x.abs ≤ -x ⊔ x

lemma EReal_max_bounds_abs {a : ℕ → EReal} :
    (fun n ↦ ↑(a n).abs) ≤ (fun n ↦ (a n) ⊔ -(a n))

lemma EReal_Tendsto_neg {a : ℕ → EReal} (h : Tendsto a atTop (nhds 0)) :
    Tendsto (fun n ↦ -a n) atTop (nhds 0)

lemma EReal_tendsto_max {a : ℕ → EReal} (h : Tendsto a atTop (nhds 0)) :
    Tendsto (fun n ↦ (a n) ⊔ -(a n)) atTop (nhds 0)
```

Some useful techniques I used to prove these are induction over the `EReal`s so that I could deal with cases of top, bot and ℝ separately and the lift tactic.

Reflection

Lastly, I set out to prove an additional corrolary but I did not reach this step. I however stated the theorem and introduced a Lesbesgue integral and some notation for it.

In this project I opened a can of worms. I found myself having to work with many translations from `EReal`s to `ENNReal`s and ℝ which was really hard. I understand why EReals do not have much API (because it is not really used anywhere) but by building some standard lemmas in Measure Theory and EReals, I gained familiarity with filters, induction on EReals, coercions and some of the lemmas for measure theory.